
4th Dimension Technical Note   #56               Transferring records from one data base to another          Page 1

#56 Transferring records from one database to another

Written By: Wesley C. Swannack January 9, 1987.
Jezreel, Inc.

Modified by: Dave Terry Feb 3, 1988
Published Date: Feb 3, 1988
_____________________________________________________________

Checking the status of the records to be duplicated while using SEND RECORD
_____________________________________________________________

Technical note #25, showed how to use the "Send Record" and "Receive Record" commands to clone an entire 
record with a minimal amount of effort.  This method can also be used when it is necessary to transfer data between 
two 4th Dimension databases that are not physically located in the same area.

The goal is to be able to clone a file's records, transfer them to disk, then load them into a second database.  As the 
records are imported we need to determine if the new record already exists, and if it does whether we should modify 
the existing record with the new information or disregard the new record.  The first example is a straight dump and 
load from one database to another.

`Dump Procedure in DatabaseA
DEFAULT FILE([Personel])
ALL RECORDS
SET CHANNEL(12;"Names")
APPLY TO SELECTION(SEND RECORD)
SET CHANNEL(11)

`Load Procedure in DatabaseB
DEFAULT FILE([PersonelArchive])
SET CHANNEL(10;"Names")
While(OK=1)

CREATE RECORD
RECEIVE RECORD
SAVE RECORD

End While
SET CHANNEL(11)

Very simple and gets the job done.  However, what if we wanted to check the record being loaded to see if it already 
exists and if it does do one of two things, disregard the new record or write over the existing record.

`Load Procedure.  Ignores new records that match old ones.
DEFAULT FILE([PeopleArchive])
SET CHANNEL(10;"Names")
While (OK=1)
CREATE RECORD
RECEIVE RECORD
If (OK=1)
xCode:=[PeopleArchive]KeyCode
PUSH RECORD
SEARCH BY INDEX([PeopleArchive]Code=xCode)
If (Records in selection=0)

POP RECORD
SAVE RECORD

Else 
POP RECORD

End if 
End if 
End while 
SET CHANNEL(11)

With this procedure we can perform actions before we just blindly stuff data into the database.  Notice that once the 
variable "xCode" has been assigned from a field of the new record, the new record is pushed onto the record stack 
(PUSH RECORD).  This allows us to search using "xCode" without losing the record just created.  In the example 
we check to see if the record exists and if it does we ignore the new one.  

The following example shows how to replace the old record with new ones.

`Load Procedure.  Replaces old records with new records.
DEFAULT FILE([PeopleArchive])
SET CHANNEL(10;"Names")
While (OK=1)

CREATE RECORD
RECEIVE RECORD
If (OK=1)

xCode:=[PeopleArchive]KeyCode
PUSH RECORD
SEARCH BY INDEX([PeopleArchive]Code=xCode)
DELETE SELECTION   `Will do nothing if no records are selected.
POP RECORD
SAVE RECORD

End if 
End while 
SET CHANNEL(11)

